

H. Gorenflo  
O. Jandl

Mesh Generation for the 20-node  
Isoparametric Solid Element by the  
Computer Program MESHGEN

H. Gorenflo  
O. Jandl

IPP 4/148 August 1977



**MAX-PLANCK-INSTITUT FÜR PLASMAFYSIK**

**8046 GARCHING BEI MÜNCHEN**

**MAX-PLANCK-INSTITUT FÜR PLASMAFYSIK**  
**GARCHING BEI MÜNCHEN**

Mesh Generation for the 20-node  
Isoparametric Solid Element by the  
Computer Program MESHGEN

H. Gorenflo  
O. Jandl

IPP 4/148

August 1977

*Die nachstehende Arbeit wurde im Rahmen des Vertrages zwischen dem  
Max-Planck-Institut für Plasmaphysik und der Europäischen Atomgemeinschaft über die  
Zusammenarbeit auf dem Gebiete der Plasmaphysik durchgeführt.*

Mesh Generation for the 20-node  
Isoparametric Solid Element  
by the Computer Program MESHGEN

J. Gorenflo  
O. Jandl

August 1977

Abstract

For finite element calculations the mesh of the structure has to be generated. This is a tedious operation when large structures with a fine subdivision into elements are involved or when the mesh generating process has to be done manually. A computer program was therefore developed which automatically generates the mesh for solid structures discretized on a rough scale by very few macroelements.

The subdivision of the macroelements into microelements, the calculation of the coordinates of the micronodes and the correlation of the nodes to the microelements are carried out automatically with the computer program.

The calculations are based on the 20-node isoparametric solid element. With the program it is possible to take void elements and elements with different material properties into account. An example is given which illustrates the calculation processes.

A printout of the program is included.

## TABLE OF CONTENTS

	<u>Page</u>
1. Mesh generating program requirements	1
2. Description of program structure	1
2.1 Isoparametric representation	1
2.2 Definition of a structure by macroelements	4
2.3 Calculation of microelements	5
2.3.1 Calculation of nodes for microelements	5
2.3.2 Description of microelements	7
2.3.3 Numbering of the nodes	8
3. Example of application	10
Figure Captions	12
4. Appendix	19
4.1 Input for MESHGEN	19
4.2 Important variables in the program	20
4.3 Output of MESHGEN	23
4.4 Input example to Fig. 3	25
4.5 List of the Program MESHGEN	27
References	36

## 1. MESH GENERATING PROGRAM REQUIREMENTS

For the static and dynamic calculation of spatial structures by means of SAP [1] the structure has to be divided into finite elements. Many structures can be sufficiently well represented by just a few large elements (macroelements) prior to deformation. Subdivision into finer elements (microelements) can be done by an automatic method. In preparing such a general mesh generating scheme, one should take the following points into account [2]:

- a) The structure should be completely represented by finite elements of the same kind.
- b) Allowance should be made for regions with different materials.
- c) The mesh should be such as to represent the real structure as closely as possible. Macroelement surfaces, which are joining surfaces are called "identical surfaces" (see Fig. 3a, macro-elements 7 and 5). For gaps in the structure one defines "void elements" (see Fig. 3a, macroelements 6, 8, 9, 11 and 12).
- d) The structure should be represented by finite elements of suitable shape which do not lead to poor numerical conditions.
- e) The system should be numbered to provide favourable calculation conditions in SAP (e.g. the bandwidth of the stiffness matrix should be a minimum).

The MESHGEN computer program was written for mesh generation meeting these requirements. It is described below.

## 2. DESCRIPTION OF PROGRAM STRUCTURE

### 2.1 Isoparametric representation

The real structure, which is represented by macroelements (hexahedron, see Fig. 1a) with 20 macro-nodes each, is entered into the MESHGEN program. This real structure is subdivided into microelements each with

20 micronodes in accordance with the requirements listed. To subdivide curvilinear bodies, it is particularly advantageous to use isoparametric elements instead of elements with straight boundaries.

The shape of the isoparametric elements is described by suitable (shape) functions (polynomials as a rule), which allows good spatial fitting to the given real structure. By means of these functions the shape of a curvilinear body is transformed to one with straight boundaries. In isoparametric representation by hexahedrons [3, 4] each real element is mapped onto a  $(\xi, \eta, \zeta)$  coordinate system with  $-1 \leq \xi \leq 1$ ,  $-1 \leq \eta \leq 1$  and  $-1 \leq \zeta \leq 1$  (see Table 1). 20 nodes according to Fig. 1a are used to define the element.

Node	Coordinates			Characteristic node values acc. to (3)
$\ell$	$\xi_\ell$	$\eta_\ell$	$\zeta_\ell$	$C_\ell$
1	1	1	1	111
2	1	1	-1	111
3	1	-1	-1	111
4	1	-1	1	111
5	-1	1	1	111
6	-1	1	-1	111
7	-1	-1	-1	111
8	-1	-1	1	111
9	1	1	0	110
10	1	0	-1	101
11	1	-1	0	110
12	1	0	1	101
13	-1	1	0	110
14	-1	0	-1	101
15	-1	-1	0	110
16	-1	0	1	101
17	0	1	1	011
18	0	1	-1	011
19	0	-1	-1	011
20	0	-1	1	011
21	0	0	0	000
22	0	0	1	001
23	0	0	-1	001
24	0	1	0	010
25	0	-1	0	010
26	1	0	0	100
27	-1	0	0	100

Table 1  $(\xi, \eta, \zeta)$  coordinates of the unit element according to Fig. 1b

The geometry of a structure is determined by the:

- fixing of the macronodes: The macronodes are given in the space (X, Y, Z) and continuously numbered (see Sect. 4.1, point 7).
- description of the macroelements: The macronodes are assigned to the continuously numbered macroelements according to certain rules (Fig. 1a) (see Sect. 4.1, point 8).
- key diagram: This determines the arrangement of the macroelements in the space ( $\xi, \eta, \zeta$ ), in which the macroelements have straight boundaries (see Figs. 2a and 3a).

By means of the shape function [4] it is possible for a macroelement with 20 nodes to calculate the (X, Y, Z) coordinates of a point ( $\xi, \eta, \zeta$ ) according to the formulae

$$\begin{aligned} X &= \sum_{\ell=1}^{20} N_{\ell} X_{\ell}, \\ Y &= \sum_{\ell=1}^{20} N_{\ell} Y_{\ell}, \\ Z &= \sum_{\ell=1}^{20} N_{\ell} Z_{\ell}. \end{aligned} \quad (1)$$

The shape functions  $N_{\ell}$  are

$$N_{\ell} = \frac{1}{8}(1+\xi\xi_{\ell})(1+\eta\eta_{\ell})(1+\zeta\zeta_{\ell})(\xi\xi_{\ell} + \eta\eta_{\ell} + \zeta\zeta_{\ell} - 2) \quad \text{for } \ell = 1(1)8,$$

$$N_{\ell} = \frac{1}{4}(1-\xi^2)(1+\eta\eta_{\ell})(1+\zeta\zeta_{\ell}) \quad \text{for } \ell = 9(2)15,$$

$$N_{\ell} = \frac{1}{4}(1-\eta^2)(1+\xi\xi_{\ell})(1+\zeta\zeta_{\ell}) \quad \text{for } \ell = 10(2)16,$$

$$N_{\ell} = \frac{1}{4}(1-\zeta^2)(1+\xi\xi_{\ell})(1+\eta\eta_{\ell}) \quad \text{for } \ell = 17(1)20.$$

The values  $\xi_{\ell}$ ,  $\eta_{\ell}$  and  $\zeta_{\ell}$  are the node coordinates of a unit element according to Table 1, and the values ( $X_{\ell}$ ,  $Y_{\ell}$ ,  $Z_{\ell}$ ) are the macronode coordinates of the real structure.

For subdivision the isoparametric macroelements corresponding to the real structure are split into the required number of microelements (the number can vary from macroelement to macroelement) in the  $\xi$ ,

$\eta$  and  $\zeta$  directions and the (X, Y, Z) coordinates of the corresponding real micronodes are determined according to eq. (1).

## 2.2 Definition of a structure by macroelements

In defining a transformed structure, one starts with the macroelement located at the bottom left of the front of the key diagram (Fig. 2a). The subsequent numbering is consecutive first in the positive  $\xi$ -direction, then in the positive  $\eta$ -direction and then in the positive  $\zeta$ -direction. "Void elements" are defined so that complicated structures can also be entered. A macroelement is said to be "void", i.e. not present, when no (X, Y, Z) coordinates are given for at least one of the nodes 1 to 8 (corner points of a macroelement) or a material with the index "0" is given. For each macroelement one can choose a material number (see Fig. 3b) which is transferred to its micro-elements. The nodes 9 to 20 (centres of the boundaries of a macroelement) are interpolated linearly (Fig. 2b) if no (X, Y, Z) coordinates are assigned to them. If one wants to subdivide a straight boundary of a real structure into microelements of varying fineness, the (X, Y, Z) coordinates of the subdividing nodes can be given outside the centre of the two adjacent nodes. Consequently, this boundary is not equally subdivided (see, for example, Fig. 3c, boundary K). In the case of real structures with curvilinear boundaries it should be ensured that the shape functions replace the boundaries of the macroelements by parabolae. The positions of the nodes 9 to 20 of a macroelement may have a strong influence on the shape of the parabolae. If this influence becomes too strong, it is advisable to use a large number of macroelements.

In defining a structure by macroelements, it is necessary to give the number of these. For this purpose we introduced the notations

- $N_\xi$  = number of macroelements in the  $\xi$ -direction,
- $N_\eta$  = number of macroelements in the  $\eta$ -direction,
- $N_\zeta$  = number of macroelements in the  $\zeta$ -direction.

The description of the macroelements is checked in the program for the correct connection between adjacent macroelements. These connections are determined by the key diagram. The suitability of

the geometric shape of a macroelement is checked by calculating its Jacobian determinant (2) at the point  $\xi = \eta = \zeta = 0$ . The value of the Jacobian determinant has to be greater than zero since the Jacobian determinant

$$[J] = \begin{vmatrix} N_{1,\xi}, N_{2,\xi}, \dots, N_{20,\xi} & x_1 & y_1 & z_1 \\ N_{1,\eta}, N_{2,\eta}, \dots, N_{20,\eta} & x_2 & y_2 & z_2 \\ N_{1,\zeta}, N_{2,\zeta}, \dots, N_{20,\zeta} & \vdots & \vdots & \vdots \\ & \vdots & \vdots & \vdots \\ & x_{20} & y_{20} & z_{20} \end{vmatrix} \quad (2)$$

with

$$N_{\ell,\xi} = \frac{\partial N_\ell}{\partial \xi} \quad \text{for } \ell = 1(1)20,$$

$$N_{\ell,\eta} = \frac{\partial N_\ell}{\partial \eta} \quad \text{for } \ell = 1(1)20,$$

$$N_{\ell,\zeta} = \frac{\partial N_\ell}{\partial \zeta} \quad \text{for } \ell = 1(1)20,$$

is the basis for calculating the volume of a finite element and therefore has to be larger than zero for every point  $(\xi, \eta, \zeta)$  within an element [5]. A value of the Jacobian determinant not larger than zero indicates an unrealistic shape of macroelement. When a macroelement is of unrealistic shape, the calculation of the micronodes and the description of the microelements are not carried out. For checking purposes the macrostructure can be plotted in  $(X, Y, Z)$  coordinates. This involves linear interpolation between the nodes so that the positions of the nodes become visible.

### 2.3 Calculation of microelements

#### 2.3.1 Calculation of nodes for microelements

To subdivide a structure, one defines the subdivision vectors  $V_\xi$ ,  $V_\eta$ , and  $V_\zeta$ , whose components give the number of microelements into which a macroelement is subdivided in the respective direction. The macroelement surfaces are oriented by introducing the indices I, J, and K (I in the  $\xi$ -direction, J in the  $\eta$ -direction and K in the

$\xi$ -direction (see Fig. 2a). If the macroelement according to Fig. 2c is regarded as the first macroelement in a structure, the numbers enumerate the microelements. The components of the subdivision vectors are then  $V_{\xi,1} = 3$ ,  $V_{\eta,1} = 2$  and  $V_{\zeta,1} = 2$ , i.e. the macroelement is subdivided into 3 microelements in the  $\xi$ -direction, 2 microelements in the  $\eta$ -direction, and 2 microelements in the  $\zeta$ -direction (making a total of 12 microelements).  $V_{\xi,I}$  is thus the number of microelements into which each I-th macroelement is subdivided in the  $\xi$ -direction. By analogy the same applies to  $V_{\eta,J}$  and  $V_{\zeta,K}$ . The number of micronodes for the microelements into which the N-th macroelement (where  $N = I + ((J-1) + (K-1) \cdot N_{\eta}) \cdot N_{\xi}$  with  $1 \leq I \leq N_{\xi}$ ,  $1 \leq J \leq N_{\eta}$  and  $1 \leq K \leq N_{\zeta}$ ), with the subdivision components  $(V_{\xi,I}, V_{\eta,J}, V_{\zeta,K})$  is subdivided is given by  $(2 \cdot V_{\xi,I} + 1) \cdot (2 \cdot V_{\eta,J} + 1) \cdot (2 \cdot V_{\zeta,K} + 1)$ , where all 27 nodes per microelement are taken into account, as shown in Fig. 1c. To enquire in the program which nodes are needed for describing the microelements, the characteristic node values C (see Table 1) is calculated according to the following formula:

$$C = 100 \cdot (k \bmod 2) + 10 \cdot (j \bmod 2) + (i \bmod 2), \quad *) \quad (3)$$

where one has

$$k = 1(1) (2 \cdot V_{\zeta,K} + 1), \quad j = 1(1) (2 \cdot V_{\eta,J} + 1) \text{ and } i = 1(1) (2 \cdot V_{\xi,I} + 1).$$

The values of  $C = 100, 010, 001$  and  $000$  indicate (see Table 1) that the node is located either in the centre of a surface of the structure ( $\ell = 22 \div 27$ ) or in the centre ( $\ell = 21$ ) of a microelement and is not required for describing a structure. The coordinates of the nodes are calculated by means of the shape functions (1). The  $\xi$ ,  $\eta$  and  $\zeta$  values are calculated with the  $k$ ,  $j$  and  $i$  values of the respective node in

\*)  $r = b \bmod m$  denotes that  $r$  is the remainder in the division  $\frac{b}{m}$ .

For  $m = 2$  one has

$$\begin{aligned} r &= 0, \text{ when } b \text{ is even,} \\ &= 1, \text{ when } b \text{ is odd.} \end{aligned}$$

The remainders of  $k$ ,  $j$  and  $i \bmod 2$  are combined in the value C as a decimal number which illustrates well the position of the corresponding node.

accordance with

$$\begin{aligned}\xi &= -1 + (i-1)/V_{\xi,I}, \quad \text{for a node of } \ell \text{ macroelement} \\ \eta &= -1 + (j-1)/V_{\eta,J}, \quad \text{for a node of } \ell \text{ macroelement} \\ \zeta &= -1 + (k-1)/V_{\zeta,K}. \quad \text{for a node of } \ell \text{ macroelement}\end{aligned}\quad (4)$$

The  $X_\ell$ ,  $Y_\ell$ ,  $Z_\ell$  values in eq. (1) are replaced by the coordinates of the respective macroelements.

### 2.3.2 Description of microelements

The number  $A$  of microelements is determined from

$$A = \sum_{N=1}^{N_\xi \cdot N_\eta \cdot N_\zeta} a_N, \quad \begin{array}{l} 0 \text{ for a "void" macroelement} \\ V_{\xi,I} \cdot V_{\eta,J} \cdot V_{\zeta,K} \text{ for "non-void" macroelements,} \end{array} \quad (5)$$

where

$$N = I + ((J-1) + (K-1) \cdot N_\eta) \cdot N_\xi$$

with  $I$  = 1(1) $N_\xi$ ,  $J$  = 1(1)  $N_\eta$  and  $K$  = 1(1)  $N_\zeta$ .

Each microelement is described by a certain configuration (Fig. 1a) of 20 nodes. The description of the microelements is compiled in the matrix  $E_{m,\ell}$  with  $\ell = 1(1)20$  and  $m = 1(1)A$  by entering the node indices for every  $m$  and  $\ell$ . As a node can be assigned to up to 8 microelements, it is advantageous to calculate each node only once and assign it to the corresponding microelements. This is done for the quantity  $C$  (3) according to Table 2. First we have to find the  $\ell$  values which are assigned to the  $C$  value. The  $m$  belonging to these  $\ell$  values is then determined to construct the matrix  $E$ . In determining  $m$  in the  $N$ -th macroelements one starts first with

$$m = 1 + (i-1)/2 + ((j-1)/2 + (k-1)/2 \cdot N_\eta) \cdot N_\xi + \sum_{n=1}^{N-1} a_n. \quad (6)$$

The combination of  $m$  and  $\ell$  is now checked for its validity according to Table 2, which shows in which direction from the node the  $m$ -th microelement has to be located. If the condition is not satisfied,  $m$  is corrected to the direction which satisfies the condition.

Finally, it is checked to see whether the new microelement  $m$  is located within the structure.

### 2.3.3 Numbering of the nodes

For the purpose of joining surfaces ("identical surfaces") which are not adjacent in the key diagram we introduced the freely selectable quantity  $\epsilon$ . Nodes which satisfy the condition

$$(x_i - x_j)^2 + (y_i - y_j)^2 + (z_i - z_j)^2 \leq \epsilon \text{ with } i \neq j \quad (7)$$

are assumed to be identical, i.e. these nodes are only stored once. For  $\varepsilon = 0$  the above condition is not checked, and so in this case several nodes with the same coordinates may exist.

The numbering of the nodes has a direct influence on the bandwidth of the stiffness matrix in the SAP program and thus governs the computing time. It is therefore advantageous to achieve optimum numbering of the nodes. This is done by numbering the nodes first in the direction with fewest nodes. For this purpose it is necessary to define the following functions:

$$\begin{aligned}
 S_1 &= 1 + 2 \cdot \sum_{I=1}^{N_\xi} V_{\xi,I} && \text{sum of the nodes in the } \xi\text{-direction} \\
 S_2 &= 1 + 2 \cdot \sum_{J=1}^{N_n} V_{n,J} && " \\
 S_3 &= 1 + 2 \cdot \sum_{K=1}^{\zeta} V_{\zeta,K} && \zeta - " \\
 Q_1 &= i + 2 \cdot \sum_{I=1}^{I_n-1} V_{\xi,I} && \begin{aligned} \text{with } i = 1(1) (2 \cdot V_{\xi, I_n+1}) \text{ and } 1 \leq I_n \leq N \\ \text{number of the } i\text{-th node in the } I_n\text{-th} \\ \text{macroelement surface in the } \xi\text{-direction} \end{aligned} \tag{8}
 \end{aligned}$$

$$Q_2 = j + 2 \sum_{J=1}^{J_n-1} v_{\eta,J}$$

with  $i = 1(1) (2 \cdot v_{\eta,J_n} + 1)$  and  $1 \leq J_n \leq N_\eta$   
number of the  $j$ -th node in the  $J_n$ -th macroelement surface in the  $\eta$ -direction (9)

$$Q_3 = k + 2 \cdot \sum_{K=1}^{K_n-1} v_{\zeta,K}$$

with  $k = 1(1) (2 \cdot v_{\zeta,K_n} + 1)$  and  $1 \leq K_n \leq N_\zeta$   
number of the  $k$ -th node in the  $K_n$ -th macroelement surface in the  $\zeta$ -direction

The function values  $S$  are checked to find the indices of  $\ell$ ,  $m$  and  $n$  for which the condition

$$S_\ell \leq S_m \leq S_n \quad (\text{with } 1 \leq \ell \leq 3, 1 \leq m \leq 3, 1 \leq n \leq 3 \text{ and } \ell \neq m \neq n)$$

is satisfied. By means of the functions  $S$  and  $Q$  and the indices  $\ell$ ,  $m$  and  $n$  a number  $K_{NR}$  is calculated for each node according to the formula

$$K_{NR} = Q_\ell + S_\ell \cdot (Q_m - 1 + S_m \cdot (Q_n - 1)). \quad (10)$$

The values of  $K_{NR}$  are between 1 and  $K_{NR \max} = S_1 \cdot S_2 \cdot S_3$ . When the nodes are arranged in ascending order of their value  $K_{NR}$ , one obtains the most favourable bandwidth. As not all of the nodes are required, the vector  $K_{NR}$  finally has to be renumbered from 1 to  $P$  ( $P$  = number of nodes required). This numbering has to be transferred to the matrix  $E$  of the description of the microelements.

Node $\ell$	Microelement in the negative $\zeta$ -direction	Microelement in the negative $\eta$ -direction	Microelement in the negative $\xi$ -direction	Value of node C
1	T	T	T	111
2	T	T	F	111
3	T	F	F	111
4	T	F	T	111
5	F	T	T	111
6	F	T	F	111
7	F	F	F	111
8	F	F	T	111
9	T	T	O	110
10	T	O	F	101
11	T	F	O	110
12	T	O	T	101
12	F	T	O	110
14	F	O	F	101
15	F	F	O	110
16	F	O	T	101

17	0	T	T	011
18	0	T	F	011
19	0	F	F	011
20	0	F	T	011

T = true, F = false, 0 = this direction not required

Table 2 Boundary conditions of the nodes

### 3. EXAMPLE OF APPLICATION

An example [2] of the possibilities available for subdividing a mesh is presented in Fig. 3. To obtain a better overall view, only one macroelement was used in the  $\eta$ -direction and Fig. 3b and c was only drawn in two dimensions for the plane  $\eta = -1$ . In the key diagram in Fig. 3a the macroelements are clearly represented in  $(\xi, \eta, \zeta)$  coordinates and used for the input points 1 to 4 (Sec. 4.1). The zone diagram in Fig. 3b, as a sketch of the structure in Cartesian coordinates, is suitable for

- a) determining the subdivision into macroelements
- b) entering the numbering of the nodes
- c) defining regions with meshes of various density by giving nodes outside the centres of the boundaries of macroelements
- d) giving "identical surfaces"
- e) assigning various materials
- f) compiling the input points 5 to 8 (Sec. 4.1).

The mesh subdivision in Fig. 3c shows the final subdivision of the real structure into microelements. One boundary of a macroelement is determined by three macronodes (two end nodes and one centre node). If the centre node is equidistant from the two end nodes, the boundary is evenly divided (Fig. 3c, boundary G). Progressive subdivision (increasingly fine division) is achieved by shifting the centre node in the direction of the finer subdivision (Fig. 3c, boundary K).

Furthermore, it can be seen in Fig. 3c that the "identical surface" of macroelements 5 and 7 have to have the same node coordinates in both macroelements. This means that the subdivision vector components  $V_{\xi,2}$  and  $V_{\xi,3}$  have to be equal ( $V_{\xi,2} = V_{\xi,3} = 4$ ) so that the micro-elements fit together.

#### ACKNOWLEDGEMENT

Die authors are grateful to Dr. Söll for many helpful suggestions during the preparation of this report.

FIGURE CAPTIONS

Fig. 1 Node numbering of a 3-dimensional solid element

- a) Basic shape of a finite element (20 nodes)
- b) Basic shape of a finite element (27 nodes)
- c) Representation in Cartesian coordinates

Fig. 2 Numbering of the macroelements

- a) Key diagram
- b) Structure
  - given nodes
  - interpolated nodes
- c) Numbering of microelements for the first macroelement

Fig. 3 Sample subdivision of a structure into finite elements

- a) Key diagram - void zones hatched
- b) Zone diagram ( $\eta = -1$ , front)
  - given nodes
  - given nodes for mesh subdivision
  - interpolated nodes
- c) Mesh subdivision ( $\eta = -1$ , front) - origin 0 ( $X=0, Z=0$ )

Fig. 4 Flowchart MAIN-Program

Fig. 5 Flowchart TRANS-Subroutine

Fig. 6 Flowchart PUNKT-Subroutine




Fig. 1







Fig. 2



a)




b)




c)

Fig. 3



MAIN-Program

Fig. 4



Subroutine TRANS

Fig. 5




Fig. 6

#### 4. APPENDIX

##### 4.1 Input for MESHGEN

1. Control card: NMKP, NMEL, NC, NR, NS (14I5)

NMKP number of macronodes

NMEL number of macroelements

NC = N<sub>ξ</sub> " " " in the ξ -direction

NR = N<sub>η</sub> " " " " " η -direction

NS = N<sub>ζ</sub> " " " " " ζ -"

2. Card for subdividing in the ξ-direction: (NUVC(I), I=1,NC) (14I5)

NUVC = V<sub>ξ,I</sub> subdivision vector of the macroelements in the ξ- direction

3. Card for subdividing in the η-direction: (NUVR(I), I=1,NR) (14I5)

NUVR = V<sub>η,J</sub> subdivision vector of the macroelements in the η- direction

4. Card for subdividing in the ζ-direction: (NUVS(I), I=1, NS) (14I5)

NUVS = V<sub>ζ</sub> subdivision vector for the macroelements in the ζ- direction

5. Card for joining node points: EPS (GE12.5)

EPS = ε smallest permissible distance between two nodes in microelements. For smaller distances the nodes are merged. For EPS = 0 the nodes are not checked for equality.

6. Material card: (MTRL(K), K=1, NMEL) (14I5)

MTRL vector of the material numbers

MTRL(K) = 0 denotes that the K macroelement is void

MTRL(K) ≠ 0 gives a material number

7. Cards for macronode coordinates: IC,(XYZM(IC,J), J=1,3) (I5,3F10.5)

IC node number (IC = 1(1) NMKP)

XYZM node coordinates X, Y and Z

(One line is entered for each node)

8. Cards for describing macroelements: IL,(IMEL(IL,L),L=1,20) (I5/(10I5)

IL macroelement number  
IMEL(IL,L) description of the macroelement by the indices of the 20 nodes ( $L = 1(1)20$ ). At  $IMEL(IL,L) = 0$  the IL-th element is void for  $L < 9$ , and for  $L > 8$  the nodes are linearly interpolated. (Three cards are entered for each macroelement).

4.2 Important variables in the program

a) COMMON /AA/ XYZ(20,3), NK1, NE1, ND1, AKSEL(3,20), IKNR, KA, JA, IA

XYZ coordinates of the particular macroelement with interpolated intermediate points  
 $NK1 = 2 V_{\xi,I}$  number of nodes in the  $\xi$ -direction  
 $NE1 = 2 V_{\eta,J}$  number of nodes in the  $\eta$ -direction  
 $ND1 = 2 V_{\zeta,K}$  number of nodes in the  $\zeta$ -direction  
AKSEL unit element  
IKNR = P number of micronodes  
KA start index within a macroelement for subdivision in the  $\xi$ -direction  
JA start index within a macroelement for subdivision in the  $\eta$ -direction  
IA start index within a macroelement for subdivision in the  $\zeta$ -direction

b) COMMON /BANDBR/ KNPNR (2000), XTOP (3,2000), MSUM (3), MMG (3), MG (3), MT (3,6), MF

KNPNR =  $K_{NR}$  micronode number  
XTOP micronode coordinates  
MSUM =  $S_i$  number of nodes in the  $\xi$ -,  $\eta$ - and  $\zeta$ -directions

MMG =  $Q_i$  node number in the  $\xi$ -,  $\eta$ - and  $\zeta$ -directions for macroelements

MG =  $Q_i$  node number in the  $\xi$ -,  $\eta$ - and  $\zeta$ -directions for microelements

MT table for numbering

MF numbering case treated

- c) COMMON /PKT/ JMEL (250,20), MELANZ (25), JC, NC, NR, NS, NUVC (10), NUVR (10), NUVS (10), MK, MJ, MI, K, J, I, KJI, MLEER (25)

JMEL =  $E_{m,\ell}$  description of microelements by indices of the 20 node coordinates ( $\ell = 1(1)20$  and  $m = 1(1)p$ )

$$MELANZ(JC) = \sum_{N=1}^{JC-1} a_N, \text{ where } a_N = \begin{cases} 0 & \text{for a void macroelement} \\ V_{\xi,I} \cdot V_{\eta,J} \cdot V_{\zeta,K} & \text{for all other macroelements} \end{cases}$$

( $I=1(1)NC$ ,  $J=1(1)NR$ ,  $K=1(1)NS$  and  $N=I+((J-1)+(K-1)\cdot N_\eta)\cdot N_\xi$ )

JC index of the actual macroelement

NC, NR, NS, NUVC, NUVR, NUVS see input data

MK, MJ, MI actual indices for macroelement

K, J, I actual indices for microelement

KJI = C characteristic value of a node

MLEER indicator for void macroelements

- d) COMMON /PLOT/ XYZM (200,3), IMEL (25,20), NMKP, NMEL

see input data

e) General quantities

INTEGER INT1(8), IM1, IM2, ID(8,3), NFG(8,2,3), KNPMAX

REAL JAC (3,3)

LOGICAL MLEER, JACDET, LORD

INT1, IM1, IM2 indices for linear interpolation of intermediate points

ID indices of the nodes required for Jacobian matrix

NFG table of indices, which are adjacent when the macroelements are properly ordered

KNPMAX =  $K_{NRmax}$  maximum number of nodes

JAC Jacobian matrix

JACDET .TRUE. if  $\det(JAC) > 0$  and .FALSE. if  $\det(JAC) \leq 0$

MLEER see COMMON /PKT/

LORD .TRUE. if the macroelements are properly ordered

f) Quantities in subroutine TRANS

REAL TN (20)

INTEGER KJIT (4)

TN shape functions

KJIT table of KJI values not required

g) Quantities in subroutine PUNKT

INTEGER MW (3,4)

LOGICAL LR 620, VR (20), GR (20)

MW	table of loop parameters for KJI
LR	logical table for left boundary
VR	logical table for front boundary
GR	logical table for base

h) Quantities in subroutine PLOTMK

INTEGER IR (7,4)

IR	table of node sequence for plotting
FRAME, SCALE, PLOTL	plot programs

### 4.3 Output of MESHGEN

1. Control printout of input data stating the maximum number of nodes (KNPMAX) and the numbering case (MT(I, MF), I=1, 3) for the micro-nodes
2. Specification of the macrostructure stating void macroelements and checking of the ordering of macroelements
3. Complete coordinates of macroelements with Jacobian matrix and determinant for  $\xi = \eta = \zeta = 0$
4. Coordinates of the micronodes with degrees of freedom. Translation is permitted ( $\Delta X \neq 0, \Delta Y \neq 0, \Delta Z \neq 0$ ), while rotation is not ( $\Delta \theta_X = \Delta \theta_Y = \Delta \theta_Z = 0$ ).
5. Description of microelements by indices of the 20 micronode coordinates stating the bandwidth and Jacobian determinant for  $\xi = \eta = \zeta \neq 0$  of the microelements for the purpose of checking.

6. Additional output of micronode coordinates with degrees of freedom and description of microelements by indices of the nodes on Logical Unit Number 7 (standard punch unit) in card image format. This output is designed for processing with the SAP program. Only the degrees of freedom for translation ( $\Delta X$ ,  $\Delta Y$ ,  $\Delta Z$ ) have still to be completed. For all nodes for which  $\Delta X = 0$ ,  $\Delta Y = 0$  or  $\Delta Z = 0$  (fixed points), the "0" has to be replaced by a "1" (see program description of SAP). The node forces required for SAP are calculated with the SHAPE program [6] from given volume forces.

The following calls in the output shold be noted:

a) \*\*\*ERROR IN THE NUMBERING\*\*\*

This message means that either the nodes (IC = 1(1)NMKP) in card 7 or the macroelements (IL = 1(1)NMEL) in card 8 of the input are not consecutively numbered.

b) \*\*\* NO GENERATION OF MICROELEMENTS \*\*\*

\*\*\* EITHER JACOBIAN DET..<= 0 OR WRONG ARRANGEMENT OF THE MACROELEMENTS \*\*\*

This message appears when the program is terminated and no micro-elements are produced.

c) \*\*\* ELEMENT - NO. .... IS A VOID ELEMENT \*\*\*

The void elements should be checked because MTRL (NO) = 0 or IMEL (NO,L) = 0 for L < 9 gives rise to this message.

d) In the table "CONNECTION OF THE MACROELEMENTS" the following message may appear alongside on the right: WRONG CONNECTION \*\*\*

This means that the numbering of the nodes is faulty and is not in keeping with the structure given in Fig. 2.

#### 4.4 Input example to Fig. 3



#### 4.5 List of the Program MESHGEN

C THIS PROGRAM SUBDIVIDES A THREE DIMENSIONAL STRUCTURE GIVEN BY  
C MACROELEMENTS INTO SMALLER MICROELEMENTS  
C JUR:F2.MESHGEN  
C NECESSARY SUBROUTINES :  
C TRANS : COMPUTATION OF THE COORDINATES OF THE MICRONODES POINTS  
C PUNKT : CORRELATION OF THE MICRONODES TO THE MACROELEMENTS  
C PLOTMK : LINEAR PLOT OF THE MACROELEMENTS  
C INPUT :  
C 1. CARD : NMKP,NMEL,NC,NR,NS (14I5)  
C NMKP = NUMBER OF MACRONODES  
C NMEL = NUMBER OF MACROELEMENTS  
C NC = NUMBER OF MACROELEMENTS IN THE KSI - DIRECTION  
C NR = NUMBER OF MACROELEMENTS IN THE ETA - DIRECTION  
C NS = NUMBER OF MACROELEMENTS IN THE ZETA - DIRECTION  
C 2. CARD : (NUVC(I),I=1,NC) (14I5)  
C NUVC = SUBDIVISION VECTOR OF THE MACROELEMENTS OF THE  
C DIRECTION KSI  
C 3. CARD : (NUVR(I),I=1,NC) (14I5)  
C NUVR = SUBDIVISION VECTOR OF THE MACROELEMENTS OF THE  
C DIRECTION ETA  
C 4. CARD : (NUVS(I),I=1,NC) (14I5)  
C NUVS = SUBDIVISION VECTOR OF THE MACROELEMENTS OF THE  
C DIRECTION ZETA  
C 5. CARD : EPS (6E12.5)  
C EPS = SMALLEST PERMISSIBLE DISTANCE BETWEEN TWO NODES.  
C FOR SMALLER DISTANCE THE NODES ARE MERGED.  
C FOR EPS=0 THE NODES ARE NOT CHECKED FOR EQUALITY.  
C 6. CARD : (MTRL(K),K=1,NMEL) (14I5)  
C MTRL = VECTOR OF THE MATERIAL NUMBERS  
C = 0 : VOID ELEMENT  
C = O : MATERIAL NUMBER  
C 7. CARD : IC,(XYZM(I,J),J=1,3) (I5,3F10.5)  
C IC = NODE NUMBER (IC=1,NMKP)  
C XYZM = NODE COORDINATES X , Y AND Z  
C ( ONE LINE IS ENTERED FOR EACH NODE )  
C 8. CARD : IL,(IMEL(JG,L),L=1,20) (I5/(10I5))  
C IL = MACROELEMENT NUMBER (IL=1,NMEL)  
C IMEL = DESCRIPTION OF THE MACROELEMENT BY THE INDICES OF THE  
C 20 NODES. FOR IMEL(IL,L)=0 AND L<9 , THE ELEMENT IL IS  
C A VOID ELEMENT. FOR IMEL(IL,L)=0 AND L>8 THE NODES ARE  
C LINEARLY INTERPOLATED.  
C ( THREE LINES ARE ENTERED FOR EACH MACROELEMENT )  
C OUTPUT :  
C 1. CONTROL PRINTOUT OF INPUT DATA STATING THE MAXIMUM NUMBER OF  
C NODES , NUMBERING CASE OF THE MICRONODES AND OF THE MATERIAL  
C NUMBERS.  
C 2. SPECIFICATION OF THE MACROSTRUCTURE STATING VOID MACRO-  
C ELEMENTS AND CHECKING OF THE ORDERING OF MACROELEMENTS.  
C 3. COMPLETE COORDINATES OF MACROELEMENTS WITH JACOBIAN MATRIX  
C AND THE DETERMINANT FOR THE POINT KSI=ETA=ZETA=0.  
C 4. COORDINATES OF THE MICROELEMENTS WITH DEGREES OF FREEDOM  
C ( TRANSLATION IS PERMITTED , ROTATION IS NOT PERMITTED )  
C 5. DESCRIPTION OF THE MICROELEMENTS BY INDICES OF THE 20 MICRO-  
C NODE COORDINATES STATING THE BANDWIDTH AND JACOBIAN-  
C DETERMINANTE FOR THE POINT KSI=ETA=ZETA=0.  
C 6. ADDITIONAL OUTPUT OF THE MICRONODES AND MICROELEMENTS TO  
//G.FT07F001 DD (DCB=(RECFM=FB,LRECL=80,BLKSIZE=2240))

```
COMMON /AA/XYZ(20,3),NK1,NE1,ND1,AKSEL(3,20),IKNR,KA,JA,IA
COMMON /BANDBR/ KNPNR(5000),XTOP(3,5000),MSUM(3),MMG(3),MG(3),
> MT(3,6),MF
COMMON/PLOT/ XYZM(200,3),IMEL(25,20),NMKP,NMEL
COMMON/PKT/JMEL(500,20),MELANZ(25),JC,NC,NR,NS,NUVC(25),NUVR(25),
> NUVS(25),MK,MJ,MI,K,J,I,KJI,MLEER(25)
INTEGER*4 INT1(8)/2,3,4,1,6,7,8,5/,ID(8,3)/10,12,14,16,17,18,19,20
>,9,11,13,15,17,18,19,20,9,10,11,12,13,14,15,16/,NCRS(3),JCIJK(3),
> MIJK(3),NFG(8,2,3)/1,12,4,20,8,16,5,17,2,10,3,19,7,14,6,18
> ,1,9,2,18,6,13,5,17,4,11,3,19,7,15,8,13
> ,1,9,2,10,3,11,4,12,5,13,6,14,7,15,8,16/,MTRL(25)/25*0/
LOGICAL*1 JACDET/.TRUE./,LORD/.TRUE./
LOGICAL MLEER,TEXT(8)/*KSI','ETA','ZETA','A','I','C','R','S*/
REAL*4 JAC(3,3)/9*0.0/
EQUIVALENCE (NCRS(1),NC),(NCRS(2),NR),(NCRS(3),NS),(MIJK(1),I1),
> (MIJK(2),J1),(MIJK(3),K1)
C READ IN OF THE MACROSTRUCTURE AND CHECKING OF THE INPUT
READ(5,100,END=2000) NMKP,NMEL,NC,NR,NS
WRITE(6,200) NMKP,NMEL,TEXT(1),TEXT(6),NC,TEXT(2),TEXT(7),NR,
> TEXT(3),TEXT(8),NS
READ(5,100) (NUVC(I),I=1,NC)
READ(5,100) (NUVR(J),J=1,NR)
READ(5,100) (NUVS(K),K=1,NS)
READ(5,103) EPS
READ(5,100) (MTRL(K),K=1,NMEL)
DO 17 I=1,NC
17 MSUM(1)=MSUM(1)+2*NUVC(I)
DO 18 J=1,NR
18 MSUM(2)=MSUM(2)+2*NUVR(J)
DO 19 K=1,NS
19 MSUM(3)=MSUM(3)+2*NUVS(K)
IF(MSUM(1).GT.MSUM(2)) MF=MF+2
IF(MSUM(1).GT.MSUM(3)) MF=MF+2
IF(MSUM(2).GT.MSUM(3)) MF=MF+1
KNPMAX=MSUM(1)*MSUM(2)*MSUM(3)
WRITE(6,201) TEXT(6),(NUVC(I),I=1,NC)
WRITE(6,201) TEXT(7),(NUVR(J),J=1,NR)
WRITE(6,201) TEXT(8),(NUVS(K),K=1,NS)
WRITE(6,210) EPS,MSUM,KNPMAX,MF,(MT(L,MF),L=1,3)
WRITE(6,123) TEXT(4)
DO 97 I=1,NMKP
READ(5,101) IC,(XYZM(I,J),J=1,3)
IF(IC.NE.I) GOTO 1000
97 WRITE(6,121) I,(XYZM(I,J),J=1,3)
WRITE(6,204) TEXT(4)
JG=0
DO 99 MK=1,NS
DO 99 MJ=1,NR
DO 99 MI=1,NC
JG=JG+1
READ(5,102,END=1000) IL,(IMEL(JG,L),L=1,20)
IF(IL.NE.JG) GOTO 1000
WRITE(6,205) IL,MTRL(IL),(IMEL(JG,L),L=1,20)
IF(MTRL(JG).EQ.0) MLEER(JG)=.TRUE.
DO 7 L=1,8
IF (IMEL(JG,L).EQ.0) MLEER(JG)=.TRUE.
7 CONTINUE
```

```
IF(JG.GT.1) MELANZ(JG)=MELANZ(JG-1)+MELIJK
MELIJK=NUVC(MI)*NUVR(MJ)*NUVS(MK)
IF(MLEER(JG)) MELIJK=0
99 CONTINUE
DO 33 JC=1,JG
IF(MLEER(JC)) WRITE(6,207) JC
33 CONTINUE
CALL PLOTMK
C   CHECKING OF THE ARRANGEMENT OF THE MACROELEMENTS
WRITE(6,208)
JC=0
DO 30 K1=1,NS
DO 30 J1=1,NR
DO 30 I1=1,NC
JC=JC+1
JCIJK(1)=JC+1
JCIJK(2)=JC+NC
JCIJK(3)=JC+NC*NR
IF(MLEER(JC)) GOTO 30
DO 32 M=1,3
LORD=.TRUE.
IF(NCRS(M).EQ.MIJK(M).OR.MLEER(JCIJK(M))) GOTO 32
WRITE(6,209) JC,(IMEL(JC,NFG(L,1,M)),L=1,8),TEXT(M),
>           JCIJK(M),(IMEL(JCIJK(M),NFG(L,2,M)),L=1,8)
DO 31 L=1,8
IF(IMEL(JC,NFG(L,1,M)).NE.IMEL(JCIJK(M),NFG(L,2,M))) LORD=.FALSE.
31 CONTINUE
IF(LORD) GOTO 32
WRITE(6,214)
JACDET=.FALSE.
32 CONTINUE
30 CONTINUE
C   COMPUTATION OF THE MACROELEMENTS
JC=0
MMG(3)=1
DO 1 MK=1,NS
MMG(2)=1
DO 15 MJ=1,NR
MMG(1)=1
DO 16 MI=1,NC
KA=MINO(MK,2)
JA=MINO(MJ,2)
IA=MINO(MI,2)
JC=JC+1
IF(MLEER(JC)) GOTO 13
20 IF(IA.EQ.1) GOTO 28
IF(MLEER(JC-1)) IA=1
28 IF(JA.EQ.1) GOTO 27
IF(MLEER(JC-NC)) JA=1
27 IF(KA.EQ.1) GOTO 23
IF(MLEER(JC-NC*NR)) KA=1
23 DO 2 IM=1,20
IF(IMEL(JC,IM).EQ.0) GOTO 14
DO 3 L=1,3
3 XYZ(IM,L)=XYZM(IMEL(JC,IM),L)
GOTO 2
14 IF(IM.GE.17) GOTO 4
```

```
IM1=IM-8
IM2=INT1(IM-8)
GOTO 5
4 IM1=IM-16
IM2=IM-12
5 DO 6 L=1,3
6 XYZ(IM,L)=0.5*(XYZM(IMEL(JC,IM1),L)+XYZM(IMEL(JC,IM2),L))
2 CONTINUE
C CALCULATION OF THE JACOBIAN-DETERMINAT FOR THE POINT KSI=ETA=ZETA=0
C ( MACROELEMENTS )
DO 21 L=1,3
DO 21 N=1,3
21 JAC(N,L)=0.0
DO 9 L=1,3
DO 9 M=1,8
DO 9 N=1,3
JAC(N,L)=JAC(N,L)-AKSEL(N,M)*XYZ(M,L)/8.+AKSEL(N, ID(M,N))*  
> XYZ(ID(M,N),L)/4.
9 CONTINUE
DETJ=JAC(1,1)*(JAC(2,2)*JAC(3,3)-JAC(2,3)*JAC(3,2))+  
> JAC(1,2)*(JAC(2,3)*JAC(3,1)-JAC(2,1)*JAC(3,3))+  
> JAC(1,3)*(JAC(2,1)*JAC(3,2)-JAC(2,2)*JAC(3,1))
IF(DETJ.LE.0.0) JACDET=.FALSE.
M=NUVC(MI)*NUVR(MJ)*NUVS(MK)
WRITE(6,211) JC,NUVC(MI),NUVR(MJ),NUVS(MK),M,(IM,(XYZ(IM,L),L  
> =1,3),IMEL(JC,IM),IM=1,20)
WRITE(6,113) JAC
WRITE(6,212) DETJ
IF(JACDET) CALL TRANS
13 MMG(1)=MMG(1)+2*NUVC(MI)
16 CONTINUE
MMG(2)=MMG(2)+2*NUVR(MJ)
15 CONTINUE
MMG(3)=MMG(3)+2*NUVS(MK)
1 CONTINUE
IF(.NOT.JACDET) GOTO 3000
C ARRANGEMENT OF THE MICRONODES
WRITE(6,123) TEXT(5)
IL=1
DO 22 M=1,KNPMAX
DO 25 I=1,IKNR
IF(-KNPNR(I).NE.M) GOTO 25
KNPNR(I)=IL
WRITE(6,121) IL,(XTOP(L,I),L=1,3)
WRITE(7,301) IL,(XTOP(L,I),L=1,3)
IA=I+1
IF(IA.GT.IKNR) GOTO 22
IF(EPS.EQ.0.0) GOTO 37
DO 36 J=IA,IKNR
IF((XTOP(1,I)-XTOP(1,J))**2+(XTOP(2,I)-XTOP(2,J))**2+(XTOP(3,I)-  
> XTOP(3,J))**2.LE.EPS**2) KNPNR(J)=IL
36 CONTINUE
37 IL=IL+1
GOTO 22
25 CONTINUE
22 CONTINUE
WRITE(6,204) TEXT(5)
```

```
JC=MELANZ(JG)+MELIJK
DO 10 IL=1,JC
IBAND=0
DO 11 L=2,20
L1=L-1
DO 11 M=1,L1
11 IBAND=MAX0(IABS(KNPNR(JMEL(IL,L))-KNPNR(JMEL(IL,M)))+1,IBAND)
C   CALCULATION OF THE JACOBIAN-DETERMINAT FOR THE POINT KSI=ETA=ZETA=0
C   ( MICROELEMENTS )
DO 34 L=1,3
DO 34 N=1,3
34 JAC(N,L)=0.0
DO 35 L=1,3
DO 35 M=1,8
DO 35 N=1,3
JAC(N,L)=JAC(N,L)-AKSEL(N,M)*XTOP(L,JMEL(IL,M))/8.0+
>           AKSEL(N, ID(M,N))*XTOP(L,JMEL(IL, ID(M,N)))/4.0
35 CONTINUE
DETJ=JAC(1,1)*(JAC(2,2)*JAC(3,3)-JAC(2,3)*JAC(3,2))+  
>           JAC(1,2)*(JAC(2,3)*JAC(3,1)-JAC(2,1)*JAC(3,3))+  
>           JAC(1,3)*(JAC(2,1)*JAC(3,2)-JAC(2,2)*JAC(3,1))
DO 38 L=1,NMEL
K=NMEL+1-L
IF(IL.GT.MELANZ(K)) GOTO 39
38 CONTINUE
39 WRITE(6,205) IL,MTRL(K),(KNPNR(JMEL(IL,L)),L=1,20),IBAND,DETJ
IF (DETJ.LE.0.0) JACDET=.FALSE.
WRITE(7,302) IL,MTRL(K),(KNPNR(JMEL(IL,L)),L=1,20)
10 CONTINUE
REWIND 7
IF (.NOT.JACDET) GOTO 3000
2000 STOP
1000 WRITE(6,206)
3000 WRITE(6,202)
      WRITE(7,202)
      STOP
100 FORMAT(14I5)
101 FORMAT(I5,3F10.5)
102 FORMAT(I5/(10I5))
103 FORMAT(6E12.5)
113 FORMAT('0JACOBIAN-MATRIX (KSI=ETA=ZETA=0)'/(1X1P3E15.6))
121 FORMAT(I5,3(4X'0'),3(4X'1'),1P3E13.5)
123 FORMAT('1COORDINATES OF THE M'A1,'CRO-NODE POINTS'/'0    I BOUNDAR  
>Y CONDITION (FREE=0,FIXED=1) X'12X'Y'12X'Z'/'9X'TRANSLATION     ROT  
>ATION'/'5X2(4X'X'4X'Y'4X'Z')/')
200 FORMAT('1NUMBER OF THE MACRONODES :''T51,'NMKP =',I5/
>           ' NUMBER OF THE MACROELEMENTS :''T51,'NMEL =',I5/
>           3(' NUMBER OF THE MACROELEMENTS IN 'A4,' -DIRECTION :'  
>           T51,'N'A1,' =',I5/))
201 FORMAT('0SUBDIVISIONVECTOR NUV'A1/(I5))
202 FORMAT('0**** NO GENERATION OF MICROELEMENTS ****'/
>           '0*** EITHER JACOBIANDETERM. <= 0 OR WRONG ARRANGEMENT OF THE  
>           MACROELEMENTS ***'/)
204 FORMAT('1DESCRIPTION OF THE M'A1,'CRO-ELEMENTS' T114,'BAND JACOBIA  
>N-DET.'/6X'MATERIAL' T113,'WIDTH KSI=ETA=ZETA=0'/' EL-NO. NO.')
205 FORMAT(23I5,1PE15.5)
206 FORMAT('0*** ERROR IN THE NUMBERING ***')
```

```
207 FORMAT('0*** ELEMENT - NO.'I5,' IS A VOID ELEMENT ***')
208 FORMAT(/'OCONNECTION OF THE MACROELEMENTS ://''ELEMENT NODE-POINT
>S - NO.' T52,'DIRECTION' T63,'ELEMENT NODE-POINTS - NO.')
209 FORMAT(/2(I5,3X,8I5,5XA4,4X))
210 FORMAT('OSMALLEST DISTANCE BETWEEN TWO NODES : EPS ='1PE12.4/
>      'ONUMBER OF NODES :''I5,' ''I5,' ''I5,' =''I5,' (27 NODES PER
> MICROELEMENT)'/
>      'ONUMBERING CASE MF ='I5,5X'NC NR NS/T29,3I5)
211 FORMAT('2'I3,'. ELEMENT SUBDIVIDED IN''I6,'*'I2,'*'I2,' =',I3,' MIC
>OELEMENTS'/
>      5X'M'4X'X'12X'Y'12X'Z'11X' NP'/(I6,1P3E13.5,I6))
212 FORMAT('OJACOBIAN-DETERMINANT (KSI=ETA=ZETA=0) ='1PE13.5)
214 FORMAT('+'T111,'WRONG CONNECTION ***')
301 FORMAT(I5,3(4X'0'),3(4X'1'),1P3E10.3)
302 FORMAT(I5,I15,T30,'1'/14I5,'**'/2I5/4I5)
      END
C   PRORGRAM FOR COMPUTING THE COORDINATES OF THE MICRONODES
      SUBROUTINE TRANS
      COMMON /AA/XYZ(20,3),NK1,NE1,ND1,AKSEL(3,20),IKNR,KA,JA,IA
      COMMON/PKT/JMEL(500,20),MELANZ(25),JC,NC,NR,NS,NUVC(25),NUVR(25),
>          NUVS(25),MK,MJ,MI,K,J,I,KJI,MLEER(25)
      COMMON /BANDBR/ KNPNR(5000),XTOP(3,5000),MSUM(3),MMG(3),MG(3),
>          MT(3,6),MF
      REAL*4 TN(20),KSI
      LOGICAL MLEER
      INTEGER KJIT(4)/100,10,1,0/
C   SUBDIVISION OF MACROELEMENTS
      NK1=2*NUVC(MI)+1
      NE1=2*NUVR(MJ)+1
      ND1=2*NUVS(MK)+1
      MG(3)=MMG(3)+KA-1
      DO 1 K=KA,ND1
      MG(2)=MMG(2)+JA-1
      ZETA=-1.0+(K-1.)/NUVS(MK)
      DO 2 J=JA,NE1
      MG(1)=MMG(1)+IA-1
      ETA=-1.0+(J-1.)/NUVR(MJ)
      DO 3 I=IA,NK1
      KSI=-1.0+(I-1.)/NUVC(MI)
C   CHECK IF MICRONODE IS NECESSARY
      KJI=100*MOD(K,2)+10*MOD(J,2)+MOD(I,2)
      DO 61 KK=1,4
      IF(KJI.EQ.KJIT(KK)) GOTO 4
61 CONTINUE
      KNPNR(IKNR+1)=-((MG(MT(3,MF))-1)*MSUM(MT(2,MF))+MG(MT(2,MF))-1)*
>          MSUM(MT(1,MF))+MG(MT(1,MF)))
      IF (IKNR.EQ.0) GOTO 41
      DO 5 KK=1,IKNR
      IF(KNPNR(KK).EQ.KNPNR(IKNR+1)) GOTO 4
5 CONTINUE
41 IKNR=IKNR+1
C   SHAPE - FUNCTION ACCORDING TO GALLAGHER
      DO 23 M=1,20
      DK=1.0+KSI*AKSEL(1,M)
      DE=1.0+ETA*AKSEL(2,M)
      DZ=1.0+ZETA*AKSEL(3,M)
      IF(M.LE.8) TN(M)=0.125*DK*DE*DZ*(DK+DE+DZ-5.0)
      IF(AKSEL(1,M).EQ.0.0) TN(M)=0.25*(1.0-KSI**2)*DE*DZ
```

```
IF(AKSEL(2,M).EQ.0.0) TN(M)=0.25*(1.0-ETA**2)*DK*DZ  
IF(AKSEL(3,M).EQ.0.0) TN(M)=0.25*(1.0-ZETA**2)*DK*DE  
23 CONTINUE  
DO 7 L=1,3  
DO 7 M=1,20  
XTOP(L,IKNR)=XTOP(L,IKNR)+XYZ(M,L)*TN(M)  
7 CONTINUE  
CALL PUNKT  
4 MG(1)=MG(1)+1  
3 CONTINUE  
MG(2)=MG(2)+1  
2 CONTINUE  
MG(3)=MG(3)+1  
1 CONTINUE  
RETURN  
END  
C CORRELATE THE MICRONODAL POINTS TO THE MICROELEMENTS  
SUBROUTINE PUNKT  
COMMON /AA/XYZ(20,3),NK1,NE1,ND1,AKSEL(3,20),IKNR,KA,JA,IA  
COMMON/PKT/JMEL(500,20),MELANZ(25),JC,NC,NR,NS,NUVC(25),NUVR(25),  
> NUVS(25),MK,MJ,MI,K,J,I,KJI,MLEER(25)  
LOGICAL MLEER  
INTEGER MW(3,4)/10,16,2,17,20,1,9,15,2,1,8,1/  
LOGICAL*1 LR(20)/T,2*T,2*T,3*T,3*T,2*T,2*T,2*T,2*T/  
LOGICAL*1 VR(20)/2*T,2*T,2*T,2*T,3*T,3*T,2*T,2*T,2*T/  
LOGICAL*1 GR(20)/4*T,4*T,4*T,4*T,4*T,4*T,4*T,4*T,4*T/  
C DETERMINE THE POSSIBLE INDICES IN THE ELEMENT - DESCRIPTION  
L=MOD(KJI-2,7)  
MA=MW(1,L)  
ME=MW(2,L)  
MS=MW(3,L)  
C SEARCH THE MICROELEMENTS FOR THE NODE  
60 DO 11 M=MA,ME,MS  
MJC1= JC  
I1=(I-1)/2  
NKS1=NUVC(MI)  
J1=(J-1)/2  
NETA=NUVR(MJ)  
K1=(K-1)/2  
NDZT=NUVS(MK)  
IF(KJI.EQ.110) GOTO 13  
C KSI - DIRECTION  
IF(LR(M)) GOTO 12  
IF(I.NE.NK1) GOTO 13  
IF(MI.EQ.NC) GOTO 11  
MJC1=MJC1+1  
I1=0  
NKS1=NUVC(MI+1)  
GOTO 13  
12 IF(I.NE.1) GOTO 13  
IF(MI.EQ.1) GOTO 11  
MJC1=MJC1-1  
NKS1=NUVC(MI-1)  
I1=NKS1  
13 IF(KJI.EQ.101) GOTO 15  
C ETA - DIRECTION  
IF(VR(M)) GOTO 14
```

```
IF(J.NE.NE1) GOTO 15
IF(MJ.EQ.NR) GOTO 11
MJC1=MJC1+NC
J1=0
NETA=NUVR(MJ+1)
GOTO 15
14 IF(J.NE.1) GOTO 15
IF(MJ.EQ.1) GOTO 11
MJC1=MJC1-NC
NETA=NUVR(MJ-1)
J1=NETA
15 IF(KJI.EQ.011) GOTO 17
C ZETA - DIRECTION
IF(GR(M)) GOTO 16
IF(K.NE.ND1) GOTO 17
IF(MK.EQ.NS) GOTO 11
MJC1=MJC1+NC*NR
K1=0
GOTO 17
16 IF(K.NE.1) GOTO 17
IF(MK.EQ.1) GOTO 11
MJC1=MJC1-NC*NR
K1=NUVS(MK-1)
17 IF(LR(M)) I1=I1-1
IF(VR(M)) J1=J1-1
IF(GR(M)) K1=K1-1
IF(MLEER(MJC1)) GOTO 11
C REGISTER THE NODES IN THE MICROELEMENT-DESCRIPTION
IM=MELANZ(MJC1)+(K1*NETA+J1)*NKS1+I1+1
JMEL(IM,M)=IKNR
11 CONTINUE
RETURN
END
C CONSTANTS AND INITIALVALUE
BLOCK DATA
COMMON /AA/XYZ(20,3),NK1,NE1,ND1,AKSEL(3,20),IKNR,KA,JA,IA
COMMON/PKT/JMEL(500,20),MELANZ(25),JC,NC,NR,NS,NUVC(25),NUVR(25),
> NUVS(25),MK,MJ,MI,K,J,I,KJI,MLEER(25)
COMMON /BANDBR/ KNPNR(5000),XTOP(3,5000),MSUM(3),MMG(3),MG(3),
> MT(3,6),MF
LOGICAL MLEER
DATA JMEL/10000*0/,MELANZ/25*0/,IKNR/0/,MF/1/,MSUM/3*1/,MMG/3*1/,
> MG/3*0/,MT/1,2,3,1,3,2,2,1,3,3,1,2,2,3,1,3,2,1/
DATA AKSEL/3*1.,-1.,2*1.,2*-1.,2*1.,-1.,3*1.,2*-1.,1.,4*-1.,1.,
> 2*-1.,0.,2*1.,-1.,0.,1.,0.,-1.,2*1.,0,1.,0,1.,2*-1.,0,-1.,0,
> 2*-1.,1.,0,-1.,2*1.,0,-1.,1.,0.,2*-1.,0,1.,-1.,0/
DATA XYZ/60*0./,KNPNR/5000*0/,XTOP/15000*0.0/,MLEER/25*.FALSE./
END
C LINEAR PLOT OF THE MACROELEMENTS
SUBROUTINE PLOTMK
COMMON/PLOT/ XYZM(200,3),IMEL(25,20),NMKP,NMEL
COMMON/PKT/JMEL(250,20),MELANZ(25),JC,NC,NR,NS,NUVC(10),NUVR(10),
> NUVS(10),MK,MJ,MI,K,J,I,KJI,MLEER(25)
LOGICAL MLEER
REAL*4 XYZMIN(3)/3*1E70/,XYZMAX(3)/3*-1E70/,XYZP(7,3)
INTEGER IR(7,4)/4,12,1,9,2,10,3,8,20,4,11,3,19,7,
> 5,16,8,15,7,14,6,1,17,5,13,6,18,2/
```

```
DO 10 M=1,NMKP
DO 10 L=1,3
XYZMIN(L)=AMIN1(XYZMIN(L),XYZM(M,L))
XYZMAX(L)=AMAX1(XYZMAX(L),XYZM(M,L))
10 CONTINUE
X1=XYZMAX(1)-XYZMIN(1)
Y1=XYZMAX(2)-XYZMIN(2)
Z1=XYZMAX(3)-XYZMIN(3)
CALL FRAME(0.,0.,1.,1.)
DO 11 M=1,NMEL
IF(MLEER(M)) GOTO 11
DO 13 K=1,4
NP=0
DO 12 I=1,7
IF(IMEL(M,IR(I,K)).EQ.0) GOTO 12
NP=NP+1
DO 14 L=1,3
14 XYZP(NP,L)=XYZM(IMEL(M,IR(I,K)),L)
12 CONTINUE
XMI=XYZMIN(1)-1.1*Y1-0.05*X1
YMI=XYZMIN(2)-0.05*Y1
XMA=XYZMIN(1)+1.1*X1
YMA=XYZMIN(2)+1.1*(Y1+Z1)
CALL SCALE(XMI,YMI,XMA,YMA)
CALL PLOTL(XYZP(1,1),XYZP(1,2),NP)
YMI=XYZMIN(3)-1.1*Y1-0.05*Z1
YMA=XYZMIN(3)+1.1*Z1
CALL SCALE(XMI,YMI,XMA,YMA)
CALL PLOTL(XYZP(1,1),XYZP(1,3),NP)
XMI=XYZMIN(2)-0.05*Y1
XMA=XYZMIN(2)+1.1*(Y1+X1)
CALL SCALE(XMI,YMI,XMA,YMA)
CALL PLOTL(XYZP(1,2),XYZP(1,3),NP)
13 CONTINUE
11 CONTINUE
RETURN
END
```

REFERENCES

- [1] J.K. Bathe, E.L. Wilson, F.E. Peterson: "SAP IV (A Structural Analysis Program for Static and Dynamic Response of Linear Systems)"; University of California, Berkeley (1974)
- [2] O.C. Zienkiewicz and D.V. Phillips: "An Automatic Mesh Generation Scheme for Plane and Curved Surfaces by Isoparametric Co-ordinates"; INT. J. NME, Vol. 3, 519-528 (1971)
- [3] J. Ergatoudis, B.M. Irons and O.C. Zienkiewicz: "Curved Iso-parametric, Quadrilateral Elements for Finite Element Analysis"; Int. J. Solids Struct. 4, 31-42 (1968)
- [4] R.H. Gallagher: "Finite Element Analysis-Fundamentals"; Prentice-Hall, Inc., New Jersey (1975)
- [5] R.D. Cook: "Concepts and Applications of Finite Element Analysis"; John Wiley & Sons, Inc., New York (1974)
- [6] H. Gorenflo, O. Jandl: "Calculation of the Nodal Forces for the 20-node Isoparametric Solid Element"; IPP Report, to be published.